Behavior Research Methods, Instruments, & Computers
1985, 17 (3), 441-442

Orthogonal polynomial coefficients and trend analysis for unequal intervals and unequal Ns: A microcomputer application

DANIEL COULOMBE
University of Ottawa, Ottawa, Ontario, Canada

If treatment levels in an experimental design are quantitative in that each level represents a point on a continuous scale, then the relationship between the treatment levels and the population means for the dependent variable can be studied. Techniques for this type of study are termed trend analysis and are discussed in most textbooks on experimental designs (Keppel, 1973; Kirk, 1982; Winer, 1971).

In order to perform trend analysis, a set of orthogonal coefficients representing the trends under study must be used. Extensive tables exist in which the required coefficients can be found (Fisher & Yates, 1957). Less extensive tables are also available in current texts on experimental design. However, these tables can be used only for studies in which the distance between each treatment level is equal and the number of subjects in each level is equal. If one or both of these conditions is (are) absent, trend analysis still can be performed, but the appropriate coefficients must be computed beforehand. Algorithms to do so are presented by Gaito (1965) and by Kirk (1982).

Current algorithms to compute orthogonal polynomials involve the solution of simultaneous equations. In Gaito (1965) and in Kirk (1982), algebraic solutions for linear and quadratic coefficient sets are available. However, solving the equations becomes very tedious for higher order polynomials. Tyson and Fieldes (1982) developed a routine to compute orthogonal polynomials when unequal intervals between treatment levels are used. Dunlap (1975) and Possamai (1975) presented FORTRAN programs that can be used to compute the coefficients appropriate for all situations. In this paper, I present a BASIC program having the capability of generating the orthogonal polynomials and performing all steps involved in trend analysis. This program can be used following an analysis of variance involving any number of treatments.

The Computer Program. The computer program is listed in Table 1. The output, a sample of which is reproduced in Table 2, includes a listing of the orthogonal coefficients, the ANOVA table with tests of significance of each trend, and tests of goodness of fit, with the polynomial equations and tests of departure from trend.

Table 1
Program Listing (TRENDAN.BAS)

```
10 CLS : DEFINT I,J,K,N
20 INPUT "Number of groups or conditions";K:PRINT:PRINT
30 DIM A(K,K),A$(K),B(K+1),C(K),C1(K,K),L(K),N(K),R3(K,K),X(K,K),XB(K),Y(K,K+1),
Z(K)
40 FOR I=1 TO K:PRINT "Group";I;" :";TAB(20) "N ===> ";:LINE INPUT N$:N(I)=VAL(N
$):NN=NN+N(I)
50 PRINT TAB(40) CHR$(30);"Mean ===> ";:LINE INPUT XB$:XB(I)=VAL(XB$):S=S+XB(I)*
N(I)
60 PRINT TAB(60) CHR$(30);"Level ===> ";:LINE INPUT L$:L(I)=VAL(L$)
70 NEXT I
80 PRINT:INPUT "Treatment Sum of Squares";R(1,1):R(1,2)=K-1:R(1,3)=R(1,1)/R(1,2)
90 INPUT "Error Sum of Squares";R(2,1):R(2,2)=NN-K:R(2,3)=R(2,1)/R(2,2):R(1,4)=R
(1,3)/R(2,3)
100 R(3,1)=R(1,1)+R(2,1):R(3,2)=NN-1
110 DATA LINEAR,QUAD,CUBIC,QUARTIC,QUINTIC,6th,7th,8th,9th,10th
120 CLS:PRINT "TREND    ";TAB(10);:FOR I=1 TO K:PRINT USING "    &##&    ";"C(";I
;")";:NEXT  I
130 PRINT:PRINT STRING$(79,205)
140 N=K:FOR I1=1 TO N:C1(1,I1)=1:NEXT I1:CC=1
150 FOR I=2 TO -(10<K)*10-(K<=10)*K:READ A$(I-1)
160 FOR I1=I-1 TO 0 STEP -1:FOR I2=1 TO K:A(I2,I-I1)=L(I2)^I1:NEXT I2,I1
170 FOR I1=1 TO K:FOR I2=1 TO I:X(I1,I2)=A(I1,I2)*N(I1):NEXT I2,I1
180 FOR I3=1 TO CC:FOR I1=1 TO I:Y(I3,I1)=0:FOR I2=1 TO K
190 Y(I3,I1)=Y(I3,I1)+C1(I3,I2)*X(I2,I1)
200 NEXT I2,I1,I3
210 FOR I1=1 TO CC:FOR I2=1 TO CC+1:R3(I1,I2)=Y(I1,I2+1):NEXT I2,I1
220 IF CC=1 THEN R3(1,1)=1/R3(1,1):GOTO 280
230 FOR I3=1 TO CC:FOR I1=1 TO CC-1:Z(I1)=R3(1,I1+1)/R3(1,1):NEXT I1
240 Z(CC)=1/R3(1,1):FOR I1=1 TO CC-1:FOR I2=1 TO CC-1
250 R3(I1,I2)=R3(I1+1,I2+1)-R3(I1+1,1)*Z(I2):NEXT I2
260 R3(I1,CC)=-R3(I1+1,1)*Z(I2):NEXT I1
270 FOR I2=1 TO CC:R3(I1,I2)=Z(I2):NEXT I2,I3
280 B(1)=1:FOR I1=1 TO CC+1:B(I1+1)=0:FOR I2=1 TO CC
290 B(I1+1)=B(I1+1)+R3(I1,I2)*(-Y(I2,1)):NEXT I2,I1
300 CC=CC+1:PRINT A$(I-1);TAB(10);:FOR I1=1 TO K:C1(CC,I1)=0:FOR I2=1 TO I
310 C1(CC,I1)=C1(CC,I1)+B(I2)*A(I1,I2):NEXT I2:PRINT USING "######.###  ";C1(CC,
I1);:NEXT I1:PRINT
320 NEXT I
330 PRINT STRING$(79,205):GOSUB 670
340 CLS:PRINT TAB(24) "ANOVA TABLE":PRINT STRING$(61,45)
350 US$="  ######.##   ####    ######.##    ###.###"
360 PRINT "SOURCE";TAB(17)"SS";TAB(25)"DF";TAB(36)"MS";TAB(47)"F";TAB(58) "P(F)"
:PRINT STRING$(61,45)
370 PRINT "BETWEEN ";:PRINT USING US$;R(1,1);R(1,2);R(1,3);R(1,4);:D1=R(1,2):D2=
R(2,2):F=R(1,4):GOSUB 610:PRINT USING "     #.####";X
380 B(0)=S/NN:FOR I1=2 TO K
390 FOR I=1 TO K:SS=SS+C1(I1,I)*XB(I)*N(I):DEN=DEN+N(I)*C1(I1,I)^2:NEXT I:B(I1-1
)=SS/DEN:SS=SS^2/DEN
400 PRINT " ";A$(I1-1);TAB(9);:PRINT USING US$;SS;1;SS;SS/R(2,3);:D1=1:F=SS/R(2,
3):GOSUB 610:PRINT USING "     #.####";X
410 SS=0:DEN=0:NEXT I1
420 PRINT "WITHIN  ";:PRINT USING US$;R(2,1);R(2,2);R(2,3):PRINT STRING$(61,45)
430 US$=LEFT$(US$,19):PRINT "TOTAL   ";:PRINT USING US$;R(3,1);R(3,2):PRINT STRI
NG$(61,45)
440 GOSUB 670
450 CLS:PRINT "HOW MANY TRENDS TO RETAIN (MAX =";K-2;") ";:INPUT MAX:IF MAX>K-2
THEN GOTO 450
460 FOR I1=1 TO MAX:CLS:SSDEP=0
470 PRINT "POLYNOMIAL EQUATIONS AND TEST OF GOODNESS OF FIT :     ";A$(I1);" TRE
ND":PRINT STRING$(79,205)
480 PRINT "LEVEL";TAB(10) "N";TAB(17) "MEAN";TAB(25) "PRED. MEAN";TAB(40) "DEP.
FROM PATTERN";TAB(61) "C(";I1;"j)":PRINT STRING$(79,45)
490 US$="  ##    ###   ####.##      ####.##          ####.##         ###.###"
500 FOR I2=1 TO K
510 C(I2)=B(0):FOR I3=1 TO I1:C(I2)=C(I2)+B(I3)*C1(I3+1,I2):NEXT I3
520 PRINT USING US$;I2;N(I2);XB(I2);C(I2);XB(I2)-C(I2);C1(I1+1,I2)
530 SSDEP=SSDEP+N(I2)*(XB(I2)-C(I2))^2:NEXT I2:PRINT STRING$(79,205):PRINT:PRINT
540 D1=K-(I1+1):MS=SSDEP/D1:F=MS/R(2,3)
550 COLOR 0,7:PRINT USING "&####.##";"MODEL ===>  Mean = ";B(0);
560 FOR I2=1 TO I1:PRINT USING "&####.##&##&";" +";B(I2);" C(";I2;"j) ";:NEXT I2
:COLOR 7,0:PRINT
570 PRINT:PRINT "SS DEP. FROM MODEL : ";SSDEP:PRINT "DEGREES OF FREEDOM : ";D1;"
,";R(2,2):PRINT "F                  : ";F
580 GOSUB 610:PRINT "P(F)               : ";X
590 GOSUB 670:NEXT I1
600 CLS:END
610 IF F<1 THEN SV=D2:T=D1:Z=1/F ELSE SV=D1:T=D2:Z=F
620 W=2/9/SV:L=2/9/T:Y=ABS((1-L)*Z^(1/3)-1+W)/SQR(L*Z^(2/3)+W)
630 IF T<4 THEN Y=Y*(1+.08*Y^4/T^3)
640 X=.5/(1+Y*(.196854+Y*(.115194+Y*(.000344+Y*.019527))))^4
650 IF F>=1 THEN 660 ELSE X=1-X
660 RETURN
670 PRINT :PRINT :PRINT "Press any key to continue";: BEEP
680 A$=INKEY$:IF A$="" THEN GOTO 680
690 RETURN
```

The program itself is divided into five parts. Lines 10-130 include the initialization and data entry procedures. The program first requests the number of groups or conditions (treatment levels). Then, for each treatment level, the user is asked to input the number of subjects (n), the group mean, and the level value of the independent variable. Finally, the program requests the treatment and the error sums of squares.

Lines 140-330 list the routine used to compute the orthogonal coefficients. The maximum order of the poly-

The author's mailing address is: School of Psychology, University of Ottawa, 275 Nicholas, Ottawa, Ontario, Canada K1N 6N5.

Table 2
Sample Run of TRENDAN

```
Number of groups or conditions? 4

Group 1  :      N ===> 8        Mean ===> 2.75     Level ===> 0
Group 2  :      N ===> 8        Mean ===> 3.5      Level ===> 1
Group 3  :      N ===> 8        Mean ===> 6.25     Level ===> 2
Group 4  :      N ===> 8        Mean ===> 9        Level ===> 3

Treatment Sum of Squares? 194.5
Error Sum of Squares? 41

TREND       C( 1)      C( 2)      C( 3)      C( 4)

LINEAR     -1.500     -0.500      0.500      1.500
QUAD        1.000     -1.000     -1.000      1.000
CUBIC      -0.300      0.900     -0.900      0.300


                         ANOVA TABLE
----------------------------------------------------------------
SOURCE          SS       DF       MS         F         P(F)
----------------------------------------------------------------
BETWEEN       194.50     3       64.83     44.276    0.0000
 LINEAR       184.90     1      184.90    126.273    0.0000
 QUAD           8.00     1        8.00      5.463    0.0253
 CUBIC          1.60     1        1.60      1.093    0.3054
WITHIN         41.00    28        1.46
----------------------------------------------------------------
TOTAL         235.50    31
----------------------------------------------------------------


POLYNOMIAL EQUATIONS AND TEST OF GOODNESS OF FIT :      LINEAR TREND

LEVEL    N      MEAN     PRED. MEAN     DEP. FROM PATTERN     C( 1 j)
-----------------------------------------------------------------------
  1      8      2.75        2.15              0.60            -1.500
  2      8      3.50        4.30             -0.80            -0.500
  3      8      6.25        6.45             -0.20             0.500
  4      8      9.00        8.60              0.40             1.500


MODEL ===> Mean =     5.38 +    2.15 C( 1j)

SS DEP. FROM MODEL :  9.600001
DEGREES OF FREEDOM :  2 , 28
F                  :  3.278049
P(F)               :  5.130608E-02

POLYNOMIAL EQUATIONS AND TEST OF GOODNESS OF FIT :      QUAD TREND

LEVEL    N      MEAN     PRED. MEAN     DEP. FROM PATTERN     C( 2 j)
-----------------------------------------------------------------------
  1      8      2.75        2.65              0.10             1.000
  2      8      3.50        3.80             -0.30            -1.000
  3      8      6.25        5.95              0.30            -1.000
  4      8      9.00        9.10             -0.10             1.000


MODEL ===> Mean =     5.38 +    2.15 C( 1j)  +    0.50 C( 2j)

SS DEP. FROM MODEL :  1.600004
DEGREES OF FREEDOM :  1 , 28
F                  :  1.092686
P(F)               :  .3054468
```

nomial is fixed at 10 in Line 150. If higher orders are desired, every occurrence of 10 should be changed to fit the new requirement. Also, additional terms should be appended to the data list in Line 110. When executed, this section also prints the computed coefficients in a tabular form (e.g., Table 2).

Lines 340-440 output the ANOVA table including each possible trend up to the maximum of 10. In each case, the F ratio is computed and the exact probability of F is printed.

Lines 450-600 output the polynomial equations and the tests of departure from trend. In this case, and for each trend selected by the user, the program prints a table in which the n_s, observed means, means predicted by the model, departure from pattern, and coefficients are presented. In addition, the program displays the equation used to compute the expected means, as well as the test of departure from the model. Table 2 includes these various parts of the program output. The sample data are from Kirk (1982, p. 140).

Finally, lines 610-690 are subroutines to compute the exact probability of F ratios (derived from Poole & Borchers, 1979) and to pause the display between pages.

Program Language and Requirements. The routine presented in this paper was developed on a Tandy 2000 computer, using Microsoft GW-BASIC. It can be run on any microcomputer using current versions of BASIC with, at most, minor syntax modifications. The memory requirement depends solely on the number of treatment levels. About 8K RAM are necessary for problems involving 10 treatment levels.

Availability. A source listing can be obtained at no cost from the author.

REFERENCES

Dunlap, W. P. (1975). A FORTRAN IV program for trend analysis with unequal intervals and unequal n_s. *Behavior Research Methods & Instrumentation*, **7**, 478-479.

Fisher, R. A., & Yates, F. (1957). *Statistical tables for biological, agricultural and medical research.* New York: Hafner.

Gaito, J. (1965). Unequal intervals and unequal n in trend analysis. *Psychological Bulletin*, **63**, 125-127.

Keppel, G. (1973). *Design and analysis: A researcher's handbook.* New York: Prentice-Hall.

Kirk, R. E. (1982). *Experimental design: procedures for the behavioral sciences.* Belmont: Brooks/Cole.

Poole, L., & Borchers, M. (1979). *Some common BASIC programs* (3rd ed.), Berkeley, CA: Osborne/McGraw-Hill.

Possamai, C. A. (1975). CORAT: A FORTRAN subroutine to calculate orthogonal coefficients for trend analysis. *Behavior Research Methods & Instrumentation*, **7**, 315-316.

Tyson, H., & Fieldes, M. A. (1982). A BASIC program for orthogonal polynomials and retrieval of regression coefficients for original model. *Computer Programs in Biomedicine*, **15**, 151-154.

Winer, B. J. (1971). *Statistical principles in experimental design.* New York: McGraw-Hill.

(Revision accepted for publication June 7, 1985.)